Simulation difference between two similar OpAmps
Theory







simulate this circuit &ndash; Schematic created using CircuitLab

The input common mode voltage is a fancy way of saying the average voltage of the two inputs:
\begin{gather}
V_{cm} = \frac{V_{-} + V_{+}}{2}
\end{gather}

When the op-amp is operated under negative feedback you may assume that \$V_{cm} = V_{-} = V_{+}\$.

The specification is there to tell you that there is some range of input voltages that you can expect the op-amp to function correctly. This is always within the supply given to the op-amp:
\begin{gather}
V_{s-}+A \le V_{cm} \le V_{s+}-B
\end{gather}
where \$A\$ and \$B\$ are non-negative numbers and properties of the op-amp. Of course, rather than giving you these two numbers directly they give a more useful quantity, which is the input common mode voltage range:
\begin{gather}
V_{cm-} \le V_{cm} \le V_{cm+}
\end{gather}
It is necessary for this condition to be satisfied if you want to expect the op-amp to function correctly.

Note that the parameters \$A\$ and \$B\$ are potentially non-linear functions of the supply voltage (and possibly other values such as output current). You have to check the datasheet to see how the common mode voltage range changes with supply voltage.

For example, take the LF412. In figures 4 and 5 it shows the common mode voltage as a function of positive and negative supply voltages (referenced against the average supply voltage). However, it simply is not specified below \$V_{s+} - V_{s-} \le 10V\$ because presumably the op-amp simply won't work correctly with less voltage.

Now let's re-draw the circuit and pick a new ground point such that you get a symmetric supply:







simulate this circuit

In your graph, you are picking \$V_{in} \in [0V, 5V]\$. This corresponds to \$V_{+} \in [-6V, -1V]\$. Looking at figure 5 of the LF412's datasheet,  the minimum common mode voltage allowed for a -6V negative supply is about -2.9V, or in the unshifted reference you're using of 3.1V.

TL;DR: PartSim's LF412 model is broken

This aside, for whatever reason Partsim's model for the LF412 is broken (most likely the pins are out of order). Luckily, you can get the model file here. Note that it needs a bit of tweaking to get it to work with partsim.

Here's the modified model:

*//////////////////////////////////////////////////////////
*LF412 LOW OFFSET, LOW DRIFT DUAL JFET INPUT OP-AMP MODEL
*//////////////////////////////////////////////////////////
*
* connections:  non-inverting input
*               |   inverting input
*               |   |   positive power supply
*               |   |   |   negative power supply
*               |   |   |   |   output
*               |   |   |   |   |
*               |   |   |   |   |
.SUBCKT LF412N  1   2  99  50  28
*
*Features:
*Fast settling time (.01%) =           2uS
*High bandwidth =                     3MHz
*High slew rate =                   10V/uS
*Low offset voltage =                  1mV
*Low supply current =                1.8mA
*NOTE: Model is for single device only and simulated
*      supply current is 1/2 of total device current.
*
IOS 2 1 25.0P
CI1 1 0 3P
CI2 2 0 3P
R1 1 3 1E12
R2 3 2 1E12
I1 99 4 1.0M
J1 5 2 4 JX
J2 6 7 4 JX
R3 5 50 650
R4 6 50 650
*Fp2=28 MHZ
C4 5 6 4.372P
I2 99 50 800UA
EOS 7 1 POLY(1) 16 49 1E-3 1
R8 99 49 80K
R9 49 50 80K
V2 99 8 2.13
D1 9 8 DX
D2 10 9 DX
V3 10 50 2.13
EH 99 98 99 49 1
G1 98 9 5 6 20E-3
R5 98 9 10MEG
VA3 9 11 0
*Fp1=18 HZ
C3 98 11 857.516P
*Fp=30 MHz
G3 98 15 9 49 1E-6
R12 98 15 1MEG
C5 98 15 5.305E-15
G4 98 16 3 49 1E-8
L2 98 17 144.7M
R13 17 16 1K
F6  99 50 VA7 1
F5  99 23 VA8 1
D5  21 23 DX
VA7 99 21 0
D6  23 99 DX
E1  99 26 99 15 1
VA8 26 27 0
R16 27 28 50
V5  28 25 0.646V
D4  25 15 DX
V4  24 28 0.646V
D3  15 24 DX
.MODEL DX D(IS=1E-15)
.MODEL JX PJF(BETA=1.183E-3 VTO=-.65 IS=50E-12)
.ENDS
*$


Here's the simulation results with the fixed model (\$V_{out}\$ is blue, \$V_{+}\$ is black):



Interestingly the output doesn't quite line up with theory; that's because I simply assumed that because figure 5 of the datasheet doesn't show the negative supply voltage vs. common mode voltage range it won't work below that. Instead, it is simply unspecified. The model just happens to work below that, though I don't know which behavior you'll actually get in real life.